Informatics 1

Functional Programming Lecture 7

Function properties

Philip Wadler

University of Edinburgh

Compare and contrast

```
sum :: [Int] -> Int
sum xs  =  foldr (+) 0 xs

  sum [1,2,3,4]
=
  foldr (+) 0 [1,2,3,4]
```

```
sum :: [Int] -> Int
sum =  foldr (+) 0

  sum [1,2,3,4]
=
  (foldr (+) 0) [1,2,3,4]
```

Sum, Product, Concat

```
sum        :: [Int] -> Int
sum        =  foldr (+) 0

product    :: [Int] -> Int
product    =  foldr (*) 1

concat     :: [[a]] -> [a]
concat     =  foldr (++) []
```

Part I

Fold, right and left

Fold right

```
foldr :: (a -> b -> b) -> b -> [a] -> b
foldr f u []        =  u
foldr f u (x:xs)    =  x `f` (foldr f v xs)

  foldr (++) [] ["abc","def","ghi"]
=
  foldr (++) [] ("abc" : ("def" : ("ghi" : [])))
=
  "abc" ++ foldr (++) [] ("def" : ("ghi" : []))
=
  "abc" ++ ("def" ++ foldr (++) [] ("ghi" : []))
=
  "abc" ++ ("def" ++ ("ghi" ++ foldr (++) [] []))
=
  "abc" ++ ("def" ++ ("ghi" ++ []))
```

Fold left

```
foldl :: (b -> a -> b) -> b -> [a] -> b
foldl f u []        =  u
foldl f u (x:xs)    =  foldl f (u `f` x) xs

  foldl (++) [] ["abc", "def","ghi"]
=
  foldl (++) [] ("abc" : ("def" : ("ghi" : [])))
=
  foldl (++) ([] ++ "abc") ("def" : ("ghi" : []))
=
  foldl (++) (([] ++ "abc") ++ "def") ("ghi" : [])
=
  foldl (++) ((([] ++ "abc") ++ "def") ++ "ghi") []
=
  ((([] ++ "abc") ++ "def") ++ "ghi")
```

Fold right, non-empty list

```
foldr1 :: (a -> a -> a) -> [a] -> a
foldr1 f [x]      =  x
foldr1 f (x:xs)   =  x `f` (foldr1 f xs)

  foldr1 (`max`) [3, 1, 4, 2]
=
  foldr1 (`max`) (3 : (1 : (4 : (2 : []))))
=
  3 `max` foldr1 (`max`) (1 : (4 : (2 : [])))
=
  3 `max` (1 `max` foldr1 (`max`) (4 : (2 : [])))
=
  3 `max` (1 `max` (4 `max` foldr1 (`max`) (2 : [])))
=
  3 `max` (1 `max` (4 `max` 2))
```

Fold left, non-empty list

```
foldl1 :: (a -> a -> a) -> [a] -> a
foldl1 f (x:xs)  =  foldl f x xs

  foldl1 (`max`) [3, 1, 4, 2]
=
  foldl1 (`max`) (3 : (1 : (4 : (2 : [])))))
=
  foldl (`max`) 3 (1 : (4 : (2 : []))))
=
  foldl (`max`) (3 `max` 1) (4 : (2 : []))
=
  foldl (`max`) ((3 `max` 1) `max` 4) (2 : [])
=
  foldl (`max`) (((3 `max` 1) `max` 4) `max` 2)[]
=
  (((3 `max` 1) `max` 4) `max` 2)
```

Part II

Append

Append

```
(++) :: [a] -> [a] -> [a]
[] ++ ys        =  ys
(x:xs) ++ ys    =  x : (xs ++ ys)

  "abc" ++ "de"
=
  ('a' : ('b' : ('c' : []))) ++ ('d' : ('e' : []))
=
  'a' : (('b' : ('c' : [])) ++ ('d' : ('e' : [])))
=
  'a' : ('b' : (('c' : []) ++ ('d' : ('e' : []))))
=
  'a' : ('b' : ('c' : ([] ++ ('d' : ('e' : [])))))
=
  'a' : ('b' : ('c' : ('d' : ('e' : []))))
=
  "abcde"
```

Append

```
(++) :: [a] -> [a] -> [a]
[] ++ ys       =  ys
(x:xs) ++ ys   =  x : (xs ++ ys)

  "abc" ++ "de"
=
  'a' : ("bc" ++ "de")
=
  'a' : ('b' : ("c" ++ "de"))
=
  'a' : ('b' : ('c' : ("" ++ "de")))
=
  'a' : ('b' : ('c' : "de"))
=
  "abcde"
```

Properties of operators

- There are a few key properties about operators: *associativity*, *identity*, *commutativity*, *distributivity*, *zero*, *idempotence*. You should know and understand these properties.

- When you meet a new operator, the first question you should ask is “Is it associative?” The second is “Does it have an identity?”

- Associativity is our friend, because it means we don’t need to worry about parentheses. The program is easier to read.

- Associativity is our friend, because it is key to writing programs that run twice as fast on dual-core machines, and a thousand times as fast on machines with a thousand cores.

Properties of append

```
prop_append_assoc :: [Int] -> [Int] -> [Int] -> Bool
prop_append_assoc xs ys zs  =
  xs ++ (ys ++ zs)  ==  (xs ++ ys) ++ zs

prop_append_ident :: [Int] -> Bool
prop_append_ident xs  =
  xs ++ [] == xs  &&  xs == [] ++ xs

prop_append_cons :: Int -> [Int] -> Bool
prop_append_cons x xs  =
  [x] ++ xs  ==  x : xs
```

Infix vs prefix notation

```
(++) :: [a] -> [a] -> [a]
[] ++ ys       =  ys
(x:xs) ++ ys   =  x : (xs ++ ys)

prop_append_assoc :: [Int] -> [Int] -> [Int] -> Bool
prop_append_assoc xs ys zs  =
  xs ++ (ys ++ zs)  ==  (xs ++ ys) ++ zs
```

vs

```
append :: [a] -> [a] -> [a]
append [] ys       =  ys
append (x:xs) ys   =  x : append xs ys

prop_append_assoc :: [Int] -> [Int] -> [Int] -> Bool
prop_append_assoc xs ys zs  =
  append xs (append ys zs)  ==  append (append xs ys) zs
```

Efficiency

```
(++) :: [a] -> [a] -> [a]
[] ++ ys        =  ys
(x:xs) ++ ys    =  x : (xs ++ ys)

  "abc" ++ "de"
=
  'a' : ("bc" ++ "de")
=
  'a' : ('b' : ("c" ++ "de"))
=
  'a' : ('b' : ('c' : ("" ++ "de")))
=
  'a' : ('b' : ('c' : "de"))
=
  "abcde"
```

Computing `xs ++ ys` takes about n steps, where n is the length of `xs`.

A useful fact

```
-- prop_sum.hs
import Test.QuickCheck

prop_sum :: Int -> Property
prop_sum n  =  n >= 0 ==>  sum [1..n]  ==  n * (n+1) `div` 2
```

```
[melchior]dts: ghci prop_sum.hs
GHCi, version 6.8.3: http://www.haskell.org/ghc/ :? for help
*Main> quickCheck prop_sum
+++ OK, passed 100 tests.
*Main>
```

Associativity and Efficiency: Left vs. Right

Let n_1, n_2, n_3, n_4 be the lengths of $\mathtt{xs}_1, \mathtt{xs}_2, \mathtt{xs}_3, \mathtt{xs}_4$.

Associated to the left

$$((\mathtt{xs}_1 \mathbin{+\!\!+} \mathtt{xs}_2) \mathbin{+\!\!+} \mathtt{xs}_3) \mathbin{+\!\!+} \mathtt{xs}_4$$

computing takes

$$n_1 + (n_1 + n_2) + (n_1 + n_2 + n_3)$$

steps. If we have m lists of length n, it takes about $m^2 n$ steps.

Associated to the right

$$\mathtt{xs}_1 \mathbin{+\!\!+} (\mathtt{xs}_2 \mathbin{+\!\!+} (\mathtt{xs}_3 \mathbin{+\!\!+} \mathtt{xs}_4))$$

computing takes

$$n_1 + n_2 + n_3$$

steps. If we have m lists of length n, it takes about mn steps. When $m = 1000$, the first is a thousand times slower than the second!

Associativity and Efficiency: Sequential vs. Parallel

Sequential:

$$((((((x_1 + x_2) + x_3) + x_4) + x_5) + x_6) + x_7) + x_8$$

Summing 8 numbers takes 7 steps. If we have m numbers it takes $m - 1$ steps.

Parallel:

$$((x_1 + x_2) + (x_3 + x_4)) + ((x_5 + x_6) + (x_7 + x_8))$$

Summing 8 numbers takes 3 steps.



If we have m numbers it takes $\log_2 m$ steps. When $m = 1000$, the first is a hundred times slower than the second!

Map-Reduce



The Overall MapReduce Word Count Process
edureka!
Input
Splitting
Mapping
Shuffling
Reducing
Final Result
K1, V1
List(K2, V2)
K2, List(V2)
List(K3, V3)
Deer Bear River
Car Car River
Deer Car Bear
Deer Bear River
Car Car River
Deer Car Bear
Deer, 1
Bear, 1
River, 1
Car, 1
Car, 1
River, 1
Deer, 1
Car, 1
Bear, 1
Bear, (1,1)
Car, (1,1,1)
Deer, (1,1)
River, (1,1)
Bear, 2
Car, 3
Deer, 2
River, 2
Bear, 2
Car, 3
Deer, 2
River, 2